Question: I want to open a web page on the same tab when the user clicks a movieclip. I'm using this method:

```
var url:String = "http://www.google.com";
var request:URLRequest = new URLRequest(url);
try {
  navigateToURL(request);
} catch (e:Error) {
  trace("Error occurred!");
}

```

But I have no idea of how to open it sending POST vars. Is that possible?

Answer: 
Yes, it is possible, by specifying POST as the method on URLRequest, and using flash.net.URLVariables for the vars. This is the example from the documentation:

<http://help.adobe.com/en_US/FlashPlatform/reference/actionscript/3/flash/net/URLRequest.html>

```
package {
    import flash.display.Sprite;
    import flash.net.navigateToURL;
    import flash.net.URLRequest;
    import flash.net.URLRequestMethod;
    import flash.net.URLVariables;

    public class URLRequest_method extends Sprite {

        public function URLRequest_method() {
            var url:String = "http://www.[yourDomain].com/application.jsp";
            var request:URLRequest = new URLRequest(url);

            var variables:URLVariables = new URLVariables();
            variables.exampleSessionId = new Date().getTime();
            variables.exampleUserLabel = "guest";
            request.data = variables;
            request.method = URLRequestMethod.POST;

            navigateToURL(request);
        }
    }
}

```
If running in Flash Player and the referenced form has no body, Flash Player automatically uses a GET operation, even if the method is set to URLRequestMethod.POST. For this reason, it is recommended to always include a "dummy" body to ensure that the correct method is used.